PML Library:
PD11 - Sigmoidal Response Models

PD11: Protocol

- Data taken from Heyes and Brown (1956) The Growth of Leaves.
- Water content is plotted vs distance. The resultant plot takes a sigmoidal shape:



PD11: Objective

- Fit several sigmoidal models to the data:
  - Logistic
  - Gompertz
  - Weibull
  - Richards
  - Morgan-Mercer-Flodin
  - Hill
- The above show the flexibility of Phoenix models with respect to parameter names and equations. For convenience, all models are parameterized by α, β, γ, δ
- Hill Model: find estimates for:
  - Alpha, E0
  - Beta, Emax
  - Gamma, EC50
  - Delta, n

Gabrielsson & Weiner, Pharmacokinetic and Pharmacodynamic Data Analysis - Concepts and Applications, 5th Edition, Swedish Pharmacology Press (2015)

PD11: Built-in sigmoidal Hill model with baseline



Gabrielsson & Weiner, Pharmacokinetic and Pharmacodynamic Data Analysis - Concepts and Applications, 5th Edition, Swedish Pharmacology Press (2015)

PD11: Sigmoidal Models: Influence of Gamma and Delta



Figure 11.4 The behavior of the *Weibull* model when the delta σ (left) and gamma γ (right) parameters are changed. Note the asymmetric characteristics of the model.

Appendix

The *Weibull* model is known for its asymmetric characteristics. Figure 11.4 demonstrates some of the features of this model. Figure 11.5 shows the absolute residuals of the *Hill* and *Weibull* models.

Gabrielsson & Weiner, Pharmacokinetic and Pharmacodynamic Data Analysis - Concepts and Applications, 5th Edition, Swedish Pharmacology Press (2015)

Logistic Model and Equation

- The Logistic model is the simplest sigmoid model.
- The lower asymptote is 0, upper asymptote is maximum Y.
- This model is symmetric about the inflection point.
- Should be used to model processes in which the point of inflection is approximately 1/2 of maximum Y.

- Equation: $Y = \dfrac{\alpha}{1+e^{\beta-\gamma\cdot x}}$
- 3 Parameters:
  - α = the curve's maximum Y value
  - β = the curve's midpoint
  - γ = the steepness of the curve

Gabrielsson & Weiner, Pharmacokinetic and Pharmacodynamic Data Analysis - Concepts and Applications, 5th Edition, Swedish Pharmacology Press (2015)

CERTARA

PD11: Logistic Model: PML Code

```
test(){
	# use covariate statment to declare custom X variable
	covariate(X)
	# model equation
	Y = Alpha / (1 + exp(Beta-Gamma*X))
	# observed Y and error model
	error(YEps = 1)
	observe(YObs(X) = Y + YEps)
	# Fixed effects parameters with initial estimates
	fixef(Alpha = c(, 50, ))
	fixef(Beta = c(, 1, ))
	fixef(Gamma = c(, 1, ))
}
```

Gabrielsson & Weiner, Pharmacokinetic and Pharmacodynamic Data Analysis - Concepts and Applications, 5th Edition, Swedish Pharmacology Press (2015)

Gompertz Model and Equation

- The Gompertz model was originally used to model human mortality rates, with an a priori assumption that a person's resistance to death decreases as age increases. Applied examples today range from actuarial science to modeling bacterial growth curves.
- The lower asymptote is 0, upper asymptote is maximum Y.
- Special case of generalized logistic function. The right-hand or future value asymptote of the function (maximum Y) is approached much more gradually by the curve than the left-hand or lower valued asymptote.
- Equation: $Y = \alpha * e^{-e^{\beta-\gamma \cdot x}}$
- 3 Parameters:
  - α = the curve's maximum Y value
  - β = growth rate b
  - γ = growth rate c

Gabrielsson & Weiner, Pharmacokinetic and Pharmacodynamic Data Analysis - Concepts and Applications, 5th Edition, Swedish Pharmacology Press (2015)

PD11: Gompertz Model: PML Code

```
test(){
    # use covariate statement to declare custom X variable
    covariate(X)
    # model equation
    Y = Alpha * exp(-exp(Beta-Gamma*X))
    # observed Y and error model
    error(YEps = 1)
    observe(YObs(X) = Y + YEps)
    # Fixed effects parameters with initial estimates
    fixef(Alpha = c(, 50, ))
    fixef(Beta = c(, 1, ))
    fixef(Gamma = c(, 1, ))
}
```

Gabrielsson & Weiner, Pharmacokinetic and Pharmacodynamic Data Analysis - Concepts and Applications, 5th Edition, Swedish Pharmacology Press (2015)

Weibull Model and Equation

- The cumulative Weibull model also known as the “stretched exponential” function. It adds a parameter, delta, that allows the inflection point to be modified.
- The upper asymptote is maximum Y, lower asymptote can be non-zero.
- Applied: often used to model particle sizes during dissolution of a solid dosage form.
- Equation: $Y = \alpha - \beta \cdot e^{-\gamma \cdot x^{\delta}}$
- 4 Parameters:
  - α = the curve’s maximum Y value (upper asymptote)
  - β = the lower asymptote
  - γ = controls x-value of the inflection point
  - δ = the steepness of the curve

Gabrielsson & Weiner, Pharmacokinetic and Pharmacodynamic Data Analysis - Concepts and Applications, 5th Edition, Swedish Pharmacology Press (2015)

PD11: Weibull Model: PML Code

```
test(){
    # use covariate statment to declare custom X variable
    covariate(X)
    # model equation
    Y = Alpha - Beta * exp(-Gamma * X^Delta)
    # observed Y with error model
    error(YEps = 1)
    observe(YObs(X) = Y + YEps)
    # Fixed effects parameters with initial estimates
    fixef(Alpha = c(, 50, ))
    fixef(Beta = c(, 20, ))
    fixef(Gamma = c(, 0.002, ))
    fixef(Delta = c(, 1, ))
}
```

Gabrielsson & Weiner, Pharmacokinetic and Pharmacodynamic Data Analysis - Concepts and Applications, 5th Edition, Swedish Pharmacology Press (2015)

Richards Model and Equation

- The Richards model is flexible for modeling asymmetric sigmoidal curves. It also uses the delta parameter to modify the inflection point.
- The upper asymptote is maximum Y, lower asymptote can be non-zero.
- Applied: growth rate curves
- Equation: $Y = \alpha/[1 + e^{\beta - \gamma \cdot x}]^{\frac{1}{\delta}}$
- 4 Parameters:
  - α = the curve's maximum Y value (upper asymptote)
  - β = the growth rate
  - γ = inflection point on the x-axis
  - δ = controls x-value of the inflection point

Gabrielsson & Weiner, Pharmacokinetic and Pharmacodynamic Data Analysis - Concepts and Applications, 5th Edition, Swedish Pharmacology Press (2015)

PD11: Richards Model: PML Code

```
test(){
        # use covariate statement to declare custom X variable
        covariate(X)
        # model equation
        Y = Alpha / ((1 + exp(Beta-Gamma*X))^(1/Delta))
        # observed Y and error model
        error(YEps = 1)
        observe(YObs(X) = Y + YEps)
        # Fixed effects parameters with initial estimates
        fixef(Alpha = c(, 10, ))
        fixef(Beta = c(, 3, ))
        fixef(Gamma = c(, 1, ))
        fixef(Delta = c(, 2, ))
}
```

Gabrielsson & Weiner, Pharmacokinetic and Pharmacodynamic Data Analysis - Concepts and Applications, 5th Edition, Swedish Pharmacology Press (2015)

CERTARA

Morgan-Mercer_Flodin Model and Equation

- The Morgan-Mercer-Flodin model was first used to model biological efficiency, for example responses to presence of nutrients.
- Allows for assymetrical growth (i.e. inflection point is not necessarily 1/2 of the maximum).
- The upper asymptote is maximum Y, lower asymptote can be non-zero.
- Applied: growth rate curves
- Equation: $Y = \frac{\beta \cdot \gamma + \alpha \cdot x^{\delta}}{\gamma + x^{\delta}}$
- 4 Parameters:
  - α = the curve's maximum Y value (upper asymptote)
  - β = the growth rate
  - γ = growth rate
  - δ = controls x-value of the inflection point

Gabrielsson & Weiner, Pharmacokinetic and Pharmacodynamic Data Analysis - Concepts and Applications, 5th Edition, Swedish Pharmacology Press (2015)

CERTARA

PD11: Morgan-Mercer-Flodin Model: PML Code

```
test(){
        # use covariate statement to declare custom X variable
        covariate(X)
        # the model equation
        Y = (Beta * Gamma + Alpha * X^Delta) / (Gamma + X^Delta)
        # observed Y and error model
        error(YEps = 1)
        observe(YObs(X) = Y + YEps)
        # Fixed effects and initial parameter estimates
        fixef(Alpha = c(, 20, ))
        fixef(Beta = c(, 1, ))
        fixef(Gamma = c(, 1000, ))
        fixef(Delta = c(, 5, ))
}
```

Gabrielsson & Weiner, Pharmacokinetic and Pharmacodynamic Data Analysis - Concepts and Applications, 5th Edition, Swedish Pharmacology Press (2015)

CERTARA

Hill Model and Equation

- The Hill model was originally used to describe kinetics of binding oxygen to hemoglobin. It is now used extensively to model pharmacodynamics.
- With parameter names such as Emax, and EC50, this is the classical sigmoid Emax model.
- An exponent, gamma, can be used to modify the inflection point. Addition of E0 allows Y-intercept to be non-zero.
- The upper asymptote is maximum Y, lower asymptote can be non-zero.
- Equation: $Y = \alpha + \frac{\beta \cdot x^{\delta}}{\gamma^{\delta} + x^{\delta}}$
- 4 Parameters:
  - α = the lower asymptote (E0)
  - β = the upper asymptote (Emax)
  - γ = the x-value of the inflection point (EC50)
  - δ = exponent controls steepness of the curve (n)

Gabrielsson & Weiner, Pharmacokinetic and Pharmacodynamic Data Analysis - Concepts and Applications, 5th Edition, Swedish Pharmacology Press (2015)

PD11: Hill Model: PML Code

```
test(){
    # use covariate statement to declare custom X variable
    covariate(X)
    # model equation
    Y = Alpha + ((Beta * X^Delta) / (Gamma^Delta + X^Delta))
    # observed Y with error model
    error(YEps = 1)
    observe(YObs(X) = Y + YEps)
    # Fixed effects parameters with initial estimates
    fixef(Alpha = c(, 50, ))
    fixef(Beta = c(, 1, ))
    fixef(Gamma = c(, 1, ))
    fixef(Delta = c(, 1, ))
}
```

Gabrielsson & Weiner, Pharmacokinetic and Pharmacodynamic Data Analysis - Concepts and Applications, 5th Edition, Swedish Pharmacology Press (2015)

PD11: Hill Model with PD Parameters: PML Code

```
test(){
    # use covariate statement to declare custom X variable
    covariate(C)
    # model equation
    Y = E0 + ((Emax * C^n) / (EC50^n + C^n))
    # observed Y with error model
    error(YEps = 1)
    observe(YObs(C) = Y + YEps)
    # Fixed effects parameters with initial estimates
    fixef(E0 = c(, 50, ))
    fixef(Emax = c(, 1, ))
    fixef(EC50 = c(, 1, ))
    fixef(n = c(, 1, ))
}
```

Gabrielsson & Weiner, Pharmacokinetic and Pharmacodynamic Data Analysis - Concepts and Applications, 5th Edition, Swedish Pharmacology Press (2015)

PD11: Hill Emax Initial Estimates

- E0 = 1, from exploratory plot at X = 0
- Emax = 20, from exploratory plot at X = 20
- EC50 = 10, from ½ Emax
- N = exponent arbitrarily set to 1 (from reduced model)

Gabrielsson & Weiner, Pharmacokinetic and Pharmacodynamic Data Analysis - Concepts and Applications, 5th Edition, Swedish Pharmacology Press (2015)

Demo

PD11: Summary

- Fit several empirical sigmoid models to data
- Show flexibility of Phoenix model with respect to:
  - Parameter Names
  - Model Equations
- Derive initial estimates
- Fit the model to the data
- Review results

PD11: Comparison Plot

- Observed and Predicted WC vs Distance for all Models:



Gabrielsson & Weiner, Pharmacokinetic and Pharmacodynamic Data Analysis - Concepts and Applications, 5th Edition, Swedish Pharmacology Press (2015)

CERTARA

PD11: Model Comparison of Thetas

- Overall Table:

	Source	LogLik	-2LL	AIC	BIC	nParm	Condition
1	Gompertz Model	-19.97077	39.94154	47.94154	50.773741	4	98.90797
2	Logistic Model	-14.6702	29.3404	37.3404	40.172601	4	45.74762
3	Weibull Model	-13.68642	27.37284	37.37284	40.913091	5	8437.5862
4	Richards Model	-13.79045	27.5809	37.5809	41.121151	5	153.08909
5	Morgan_Mercer	-14.86061	29.72122	39.72122	43.261471	5	155736.08
6	Hill Model	-14.86061	29.72122	39.72122	43.261471	5	11.22263

- Theta Table:

	Source											
	Gompertz Model		Hill Model		Logistic Model		Morgan_Mercer_Flodin Model		Richards Model		Weibull Model	
Parameter	Estimate	CV%	Estimate	CV%	Estimate	CV%	Estimate	CV%	Estimate	CV%	Estimate	CV%
Alpha	22.5	3.31	1.65	23.33	21.5	1.72	22.1	2.51	21.2	1.77	21.1	1.55
Beta	2.11	11.35	20.4	3.76	3.96	6.15	1.65	23.33	5.69	24.81	19.8	2.69
Delta			4.56	10.49			4.56	10.50	1.62	31.76	3.18	7.74
Gamma	0.388	11.45	6.63	2.22	0.622	6.58	5590	89.60	0.777	17.89	0.00177	48.68
stdev0	0.916	18.26	0.652	18.26	0.643	18.26	0.652	18.26	0.607	18.26	0.603	18.26

Gabrielsson & Weiner, Pharmacokinetic and Pharmacodynamic Data Analysis - Concepts and Applications, 5th Edition, Swedish Pharmacology Press (2015)

PD11: Hill Model Results

- Fit



- PK Parameter Estimates

	Parameter	Estimate	Units	Stderr	CV%	2.5% CI	97.5% CI	Var. Inf. factor
1	E0	1.65308		0.38572011	23.33342	0.79364128	2.5125187	0.32608
2	Emax	20.4241		0.76819268	3.761207	18.712459	22.135741	1.2914
3	EC50	6.63298		0.14707481	2.2173263	6.3052766	6.9606834	0.053055
4	n	4.56014		0.47857079	10.494651	3.4938169	5.6264631	0.47168
5	stdev0	0.651661		0.11897479	18.257159	0.38656842	0.91675358	

Gabrielsson & Weiner, Pharmacokinetic and Pharmacodynamic Data Analysis - Concepts and Applications, 5th Edition, Swedish Pharmacology Press (2015)

PD11: Weibull Model Results

- Fit



- PK Parameter Estimates

	Parameter	Estimate	Units	Stderr	CV%	2.5% CI	97.5% CI	Var. Inf. factor
1	Alpha	21.1036		0.32729192	1.5508819	20.374348	21.832852	0.29006
2	Beta	19.8147		0.53376961	2.6938062	18.625386	21.004014	0.77607
3	Gamma	0.00177079		0.00086204988	48.681655	-0.00014997854	0.0036915585	2.0189E-06
4	Delta	3.1796		0.24614224	7.7412956	2.6311604	3.7280396	0.16369
5	stdev0	0.602595		0.11001818	18.2574	0.357459	0.847731	

Gabrielsson & Weiner, Pharmacokinetic and Pharmacodynamic Data Analysis - Concepts and Applications, 5th Edition, Swedish Pharmacology Press (2015)

Questions?

PML School: Materials

- Each model will be made available in Certara Forum
  - Link to live webinar and presentation slides
    - https://support.certara.com/forums/forum/34-pml-school/
  - Model text as file download
    - Can be imported into Phoenix model object to be run on a new dataset
  - Questions and comments can be exchanged in the Forum
    - Or can be entered into the Certara Support portal at:
      - https://support.certara.com/support
    - Or can be sent as emails to support@certara.com

Certara University

- A wide range of On Demand and Classroom courses are available through Certara University
  - Introductory, intermediate and advanced instruction in Phoenix WinNonlin, Population Modeling using NLME, IVIVC Toolkit
  - Fundamentals of Pharmacokinetics and Pharmacodynamics
  - Noncompartmental data analysis
  - Programming Bootcamp
  - Partner Lectures and Webinar series

- Please visit our Certara University web site for more information

Coming up…

Analysis of a Tissue Growth/Kill Model

Analyze a tumor cell kill model after acute dosing

April 13, 2017 | 10am EST

Presenter: Bernd Wendt

CERTARA